

Knowledge Graphs

Lecture 3 - Querying RDFS with SPARQL

3.3 More Complex SPARQL Queries

Prof. Dr. Harald Sack & Dr. Mehwish Alam
FIZ Karlsruhe - Leibniz Institute for Information Infrastructure
AIFB - Karlsruhe Institute of Technology
Autumn 2020

Karlsruher Institut für Technologie

FIZ Karlsruhe
Leibniz-Institut für Informationsinfrastruktur

Knowledge Graphs

Lecture 3: Querying RDF(S) with SPARQL

The Semantic Web Technology Stack
(not a piece of cake...)



USE **SPARQL** and QUERY ON

[1,2]

SPARQL Federated Queries

- Example: Which Mountaineers died on Mount Everest ordered by their death date?



SPARQL Federated Queries

- SPARQL enables federated queries over several RDF datasets or SPARQL endpoints via the **SERVICE** objective

```
PREFIX dct:  <http://purl.org/dc/terms/>
PREFIX dbc:  <http://dbpedia.org/resource/Category:>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX owl: <http://www.w3.org/2002/07/owl#>

 SELECT DISTINCT  ?wditemLabel ?date WHERE {
    SERVICE <http://dbpedia.org/sparql> {
       ?item dct:subject dbc:Mountaineering_deaths_on_Mount_Everest ;
             owl:sameAs ?wditem FILTER regex (?wditem, "wikidata.org") .
    }
    SERVICE <https://query.wikidata.org/sparql>{
        ?wditem wdt:P570 ?date .
        OPTIONAL {?wditem wdt:P18 ?image . }
        ?wditem rdfs:label ?wditemLabel FILTER(LANG(?wditemLabel)="en").
    }
}
ORDER BY ?date
```

- Example: connect the DBpedia with Wikidata "Which Mountaineers died on Mount Everest ordered by their death date?"
- only possible, if SPARQL endpoints permit federation

query SPARQL endpoint

- Which Mountaineers died on Mount Everest ordered by their death date?

Wikidata Query Service | Examples | Help | More tools

```
#defaultView:Timeline
PREFIX dct:  <http://purl.org/dc/terms/>
PREFIX dbc:  <http://dbpedia.org/resource/Category:>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX owl: <http://www.w3.org/2002/07/owl#>

 SELECT DISTINCT  ?wditemLabel ?date ?image WHERE {
    SERVICE <http://dbpedia.org/sparql> {
       ?item dct:subject dbc:Mountaineering_deaths_on_Mount_Everest ;
             owl:sameAs ?wditem FILTER regex (?wditem, "wikidata.org") .
    }
    SERVICE <https://query.wikidata.org/sparql> {
      ?wditem wdt:P570 ?date .
      OPTIONAL {?wditem wdt:P18 ?image . }
       ?wditem rdfs:label ?wditemLabel FILTER(LANG(?wditemLabel)="en").
    }
}
ORDER BY ?date
```

Timeline | 42 results in 821 ms | Code | Download



query SPARQL endpoint

SPARQL Variable Assignment

- Example: Select authors with their notable works and **date of publication ordered by year**

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT ?authorLabel ?bookLabel ?book ?author ?year
WHERE {
     ?author wdt:P106 wd:Q36180 ;
             wdt:P800 ?book .
     ?book   wdt:P577 ?date .
  BIND (YEAR(?date) AS ?year) FILTER (BOUND(?year))
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en"
}
} ORDER BY ?year
```

Binding a new variable

- The **BIND** form allows a value to be assigned to a variable

query SPARQL endpoint

- Example: Select authors with their notable works and **date of publication ordered by year**

Wikidata Query Service | Examples | Help | More tools | English

```
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT ?authorLabel ?bookLabel ?book ?author ?year
WHERE {
    ?author wdt:P106 wd:Q36180 ;
            wdt:P800 ?book .
    ?book   wdt:P577 ?date .
  BIND (YEAR(?date) AS ?year) FILTER (BOUND(?year))
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
} ORDER BY ?year
```

9030 results in 13210 ms | Code | Download | Link | Search

authorLabel	bookLabel	book	author	year
Hesiod	Theogony	wd:Q156498	wd:Q44233	-700
Antimachus of Teos	Epigoni (epic)	wd:Q2067424	wd:Q577773	-600
Euclid	Elements	wd:Q172891	wd:Q8747	-300
Cato the Elder	De Agri Cultura	wd:Q1180565	wd:Q180081	-160
Cicero	De re publica	wd:Q656161	wd:Q1541	-52
Cicero	De Officiis	wd:Q1180721	wd:Q1541	-43
Sappho	Ode to Aphrodite	wd:Q21070481	wd:Q17892	-5
Titus Livius	Ab urbe condita libri	wd:Q1155892	wd:Q2039	10
Seneca	De Vita Beata	wd:Q1180753	wd:Q2054	58
Pliny the Elder	Natural History	wd:Q442	wd:Q82778	74
Quintus Curtius Rufus	Histories of Alexander the Great	wd:Q27869231	wd:Q6959	100

WIKIDATA query SPARQL endpoint

SPARQL Aggregate Functions

- Example: How many authors are there and how many notable works?

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT (COUNT(?book) AS ?bookcount)
       (COUNT(DISTINCT(?author)) AS ?authorcount)
WHERE {
     ?author wdt:P106 wd:Q36180 ;
             wdt:P800 ?book .
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}
```

aggregate functions

- **COUNT** is a SPARQL **aggregate function** which counts the number of times a given expression has a bound
- More aggregate functions:
  - **SUM**
  - **AVG**
  - **MIN / MAX**
  - **SAMPLE**

query SPARQL endpoint

SPARQL Aggregate Functions

- Example: How many authors are there and how many notable works?



WIKIDATA

query SPARQL endpoint

SPARQL Aggregate Functions

- Example: which author wrote how many notable works?

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT ?authorLabel (COUNT(?book) AS ?bookcount)
WHERE {
  ?author wdt:P106 wd:Q36180 ;
          wdt:P800 ?book .
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
} GROUP BY ?authorLabel
ORDER BY DESC (?bookcount)
```

aggregate function

- The solution can be divided into groups via **GROUP BY**
- The aggregate function then is calculated for each group

query SPARQL endpoint

- Example: **which author** wrote **how many** notable works?

Wikidata Query Service | Examples | Help | More tools | English

```
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT ?authorLabel (COUNT(?book) AS ?bookcount)
WHERE {
    ?author wdt:P106 wd:Q36180 ;
            wdt:P800 ?book .
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
} GROUP BY ?authorLabel
ORDER BY DESC (?bookcount)
```

6971 results in 8323 ms | Code | Download | Link | Search

authorLabel	bookcount
Thomas Mann	57
Enid Blyton	55
Marion Zimmer Bradley	46
Elmore Leonard	46
Karel Čapek	36
Stephen King	36
Kenji Miyazawa	36
Henry Wadsworth Longfellow	34
Noam Chomsky	34
Heinrich Böll	31
Beverly Cleary	29
Boris Vian	28
Beatrix Potter	26
Judy Blume	25

query SPARQL endpoint

SPARQL Aggregate Functions

- SPARQL 1.1 provides more aggregate functions
  - `SUM`
  - `AVG`
  - `MIN`
  - `MAX`
  - `SAMPLE` – „pick“ one non-deterministically
  - `GROUP_CONCAT` – concatenate values with a designated string separator



Next Lecture...

Karlsruher Institut für Technologie
FIZ Karlsruhe
Leibniz Institute for Information Infrastructure

Picture References:

- [1] Benjamin Nowack, *The Semantic Web - Not a Piece of cake…*, at bnode.org, 2009-07-08 , [CC BY 3.0]
  http://bnode.org/blog/2009/07/08/the-semantic-web-not-a-piece-of-cake
- [2] British Crown vector illustration, publicdomainvectors.org, [Public Domain]
  https://publicdomainvectors.org/en/free-clipart/British-Crown-vector-illustration/12150.html
- [3] WIkidata Logo, Planemad, [Public Domain]
  https://commons.wikimedia.org/wiki/File:Wikidata-logo-en.svg
- [4] Ernst Haeckel, Kunstformen der Natur (1904), plate 66: Arachnida, [Public Domain]
  https://commons.wikimedia.org/wiki/File:Haeckel_Arachnida.jpg